



HL7 FHIR DevDays 2017



Clinical Quality Improvement with FHIR

Bryn Rhodes, CTO HarmonIQ Health Systems Corporation



Amsterdam, 15-17 November | [@fhir_furore](#) | [#fhirdevdays17](#) | [www.fhirdevdays.com](#)

Quality Measurement with FHIR

- What is a Quality Measure
 - A quantitative tool to assess the performance of an individual or organization related to a specific process or outcome
- An eMeasure is the digital, computable representation of a quality measure
- Typically evaluated at a population level, but can be evaluated at the individual patient level as well

Measure Types

- **Process** - A measure that focuses on ensuring a particular process is followed
 - e.g. A measure that evaluates cancer screening rates
- **Outcome** - A measure that focuses on a particular outcome
 - e.g. A measure that evaluates improvement in asthma control
- **Structure** - A measure that focuses on a health care provider's capacity or systems
 - e.g. A measure that evaluates the ratio of staff to providers
- **Patient-reported Outcome** - A measure that focuses on patient-report outcomes
 - e.g. A patient experience measure
- **Composite** - A measure that is made up of multiple measures and evaluated as a combined score


Measure Scoring Types

- **Proportion** – A measure that is calculated by determining the proportion of members of a population that meet the criteria
 - Proportion of patients that have been appropriately screened for breast cancer
- **Ratio** – Evaluates the ratio of one population to another
 - Same as a proportion, except that the “numerator” and “denominator” may come from different populations
- **Continuous Variable** – Evaluates a particular variable for every member of a population
 - Median Time from Admission To Discharge for an ER Visit

Population Criteria Types

- Initial Population
- Denominator
 - Denominator Exclusions
 - Denominator Exceptions
- Numerator
 - Numerator Exclusions

NOTE: This diagram depicts proportion measure population relationships. Population relationships are different for each scoring.



Patient-Based vs Episode-of-Care Measures

- Patient-based
 - The individual members of the population are patients
 - Criteria are expressed as true/false conditions, whether a patient is in/our of each population (initial, denominator, numerator, etc)
- Episode-of-care based
 - The individual cases in the population are encounters, or procedures
 - Criteria are expressed as lists of cases
- Key take-away is that the expressions in a particular measure all return the same type, and that type is determined by the measure scoring and whether it is a patient- or non-patient-based

Components of a measure

- Metadata
 - Identifiers, narrative description, supporting evidence, topics, governance, relationship to other artifacts, etc.
- Structure
 - The population criteria of the measure (i.e. initial population, numerator, denominator, etc)
- Logic
 - The conditions involved in building the criteria

CQL Declarations

```
library BCS_FHIR version '1.0.0'  
  
using FHIR version '3.0.0'  
  
valueset "Absence of Left Breast Value Set": 'urn:oid:2.16.840.1.113883.3.464.1004.1329.1' // ICD10CM  
valueset "Absence of Right Breast Value Set": 'urn:oid:2.16.840.1.113883.3.464.1004.1330.2' // ICD10CM  
valueset "Bilateral Mastectomy Value Set": 'urn:oid:2.16.840.1.113883.3.464.1004.1042' // Grouping  
valueset "Mammography Value Set": 'urn:oid:2.16.840.1.113883.3.464.1004.1168' // Grouping  
.....  
  
parameter "Measurement Period" Interval<DateTime>  
  
context Patient
```



Initial Population

```
define "Initial Population":  
  "Is Female"  
  and "Is Age 52 to 74 at End"  
  
define "Is Female":  
  Patient.gender = 'female'  
  
define "Is Age 52 to 74 at End":  
  AgeInYearsAt(end of "Measurement Period") between 52 and 74
```

Denominator/Numerator

```
define "Denominator":  
  "Initial Population"  
  
define "Numerator":  
  "Initial Population"  
  and "Has Mammogram In Last 36 Months"  
  
define "Has Mammogram In Last 36 Months":  
  exists(  
    [Procedure: "Mammography Value Set"] Proc  
    where Proc.status = 'completed'  
    and Proc.performed during Interval[start of "Measurement Period" - 2 years, end of "Measurement Period"]  
  )
```

The “Retrieve” and DataRequirements



Retrieve “Procedures” for this patient that have a code that is a member of the “Mammography Value Set”

```

"dataRequirement": [
  {
    "type": "Procedure",
    "codeFilter": [
      {
        "path": "code",
        "valueSetString": "Mammography Value Set"
      }
    ]
  }
],

```

Dynamic Data Requirements

```
[Procedure: "Mammography Value Set"] Proc
  where Proc.status = 'completed'
  and Proc.performed during Interval[start of "Measurement Period" - 2 years, end of "Measurement Period"]
```


```
[base]/Measure/bcs/$data-requirements&periodStart=2016&periodEnd=2016
```

```
"dataRequirement": [
  {
    "type": "Procedure",
    "codeFilter": [
      {
        "path": "code",
        "valueSetString": "Mammography Value Set"
      }
    ],
    "dateFilter": [
      {
        "path": "performed",
        "valuePeriod": {
          "start": "2014-01-01",
          "end": "2016-12-31"
        }
      }
    ]
  }
]
```


Using Profiles to Communicate Requirements

```
[Procedure: "Mammography Value Set"] Proc
  where Proc.status = 'completed'
  and Proc.performed during Interval[start of "Measurement Period" - 2 years, end of "Measurement Period"]
```

Procedure resource:




HEDIS Procedure Profile:



MeasureReport

- Individual – Evaluates the measure for a given patient
- Summary – Evaluates the measure for a population
- Patient-list – Evaluates for a population, but returns as a bundle of individual measure reports



Can we do better?

```
define "BRACA Mutation Procedures":  
  [Procedure: "BRACA1 or BRACA2 Mutation Test"] P  
    where P.status = 'completed'  
    and P.outcome = 'positive finding'  
  
define "BRACA Mutation Observations":  
  [Observation: "BRACA1 or BRACA2 Mutation"] O  
    where O.status in { 'final', 'amended' }  
  
define "History of BRACA Mutation":  
  [FamilyMemberHistory] F  
    where F.status = 'completed'  
    and F.relationship in { 'CHILD', 'PRN', 'SIB' } // NOTE: CHILD here includes adopted, should be excluded  
    and ( F.condition.code in "BRACA Mutation Observations"  
    or F.condition.code in "BRACA Mutation Procedures"  
    )
```

Refined BCS Measure

```
define "Initial Population":  
  Common."Is Female"  
  and ("Is Age 52 to 74 at End"  
    or "Is Age 42 to 74 at End and Has BRACA Mutation Results"  
  )  
  
define "Denominator":  
  "Initial Population"  
  
define "Numerator":  
  Common."Has Mammogram In Last 36 Months"  
  
define "Denominator Exclusion":  
  (Common."Is Lacking Both Breasts"  
    or (Common."Is Lacking Left Breast"  
      and Common."Is Lacking Right Breast"))
```


BCS Decision Support

```
library BCS_CDS_Logic version '0.0.0'

using FHIR version '3.0.0'

include BCS_Common_Logic version '0.0.0' called Common

context Patient

// NOTE: Using 50 to make the decision support prospective
define "Needs Screening":
  (AgeInYears() >= 50 and AgeInYears() < 75
   or (AgeInYears() >= 40 and AgeInYears() < 75 and Common."Has BRACA Mutation Indicators"))
  and not (
    (Common."Is Lacking Both Breasts"
     or (Common."Is Lacking Left Breast"
        and Common."Is Lacking Right Breast"))
  )
  and not(Common."Has Mammogram in Last 36 Months") // Need to back off 3 months to allow for scheduling...
```

Resources

CDC Opioid Prescribing Support Implementation Guide:

- <http://build.fhir.org/ig/cqframework/opioid-cds/>

HEDIS FHIR Implementation Guide:

- <http://build.fhir.org/ig/cqframework/hedis-ig/>

Publicly Available Clinical Reasoning Test Server (CQF Ruler) URL:

- <http://measure.eval.kanvix.com/cqf-ruler/baseDstu3>

Publicly Available CDS Hooks Server URL:

- <http://measure.eval.kanvix.com/cqf-ruler/cds-services>

Walkthroughs

Activity Definition \$apply

- [https://github.com/DBCG/cqf-ruler/wiki/ActivityDefinition-\\$apply-Operation](https://github.com/DBCG/cqf-ruler/wiki/ActivityDefinition-$apply-Operation)

Plan Definition \$apply

- [https://github.com/DBCG/cqf-ruler/wiki/PlanDefinition-\\$apply-Operation](https://github.com/DBCG/cqf-ruler/wiki/PlanDefinition-$apply-Operation)

CDS Hooks Request:

- <https://github.com/DBCG/cqf-ruler/wiki/CDS-Hooks-Request-Processing>

Quality Measure Evaluation:

- <https://github.com/DBCG/cqf-ruler/wiki/Quality-Measure-Processing>

Questions?